

Objectifs

- Utiliser OCaml pour manipuler des entiers et des flottants.
- Comprendre les déclarations, variables et fonctions.
- Écrire un premier programme simple.

Une donnée structurée est constituée d'un assemblage de valeurs. Des exemples classiques de telles données sont les dates ou les fiches (contacts téléphoniques, etc.).

On a trois types de données structurées en OCaml :

- les types produits (paires et n -uplets),
- les types produits nommés (enregistrements)
- les types sommes (énumérations et sommes disjointes).

1 Paires et n -uplets

Paires

Une paire est constituée de l'assemblage de deux expressions `<expr1>` et `<expr2>`. Pour construire une paire, on écrit `(<expr1>, <expr2>)`.

Exemple

```
1 # let p1 = (1, 2) ;;  
2 val p1 : int * int = (1, 2)
```

Le type d'une paire est un produit noté $\tau_1 * \tau_2$, où τ_1 et τ_2 sont les types respectifs des expressions `<expr1>` et `<expr2>`.

Les types des expressions qui constituent une paire peuvent être différents.

Exemple

```
1 # let p2 = ('a', 2.7 +. 2.2) ;;  
2 val p : char * float = ('a', 4.9)
```

Une paire peut également contenir d'autres paires.

Exemple

```
1 # let p3 = (p1, p2) ;;
2 val p3 : (int * int) * (char * float) = ((1, 2), ('a', 4.9))
```

Accès aux composantes

Une première façon d'accéder aux composantes d'une paire est d'appeler l'une des deux fonctions prédéfinies `fst` ou `snd`, qui permettent respectivement d'extraire la composante de gauche et la composante de droite d'une paire.

Exemple

```
1 # let x = fst p1 ;;
2 val x : int = 1
3
4 # let y = snd p1 ;;
5 val y : int = 2
```

La deuxième solution consiste à utiliser `let` avec un motif `(x, y)` comme ci-dessous :

Exemple

```
1 # let (x, y) = p1 ;;
2 val x : int = 1
3 val y : int = 2
```

Cette forme de déclaration permet de déconstruire une paire, en donnant un nom à chaque composante. Le motif peut être aussi complexe que n'importe quelle construction de paire.

Exemple

```
1 # let ((x, y), z) = p3 ;;
2 val x : int = 1
3 val y : int = 2
4 val z : char * float = ('a', 4.9)
```

Il est parfois utile de déconstruire une paire pour ne récupérer que quelques composantes, en ignorant les autres. Pour ne pas introduire de noms de variables inutiles, on peut utiliser le symbole `_`.

Exemple

```
1 # let ((_ , y), _) = p3 ;;
2 val y : int = 2
```

Les n -uplets

Pour regrouper un nombre quelconque de valeurs, on utilise des n -uplets (`<expr1>`, ..., `<exprn>`). Le type d'un n -uplet est un produit $\tau_1 * \dots * \tau_n$ et, comme pour les paires, les types τ_i sont quelconques.

Exemple

```
1 # let t = ('a', 1.2, (true, 0)) ;;
2 val t : char * float * (bool * int) = ('a', 1.2, (true, 0))
```

L'accès aux éléments d'un n -uplet se fait en utilisant la construction `let` avec motifs.

Exemple

```
1 # let (x, _, (_, y)) = t ;;
2 val x : char = 'a'
3 val y : int = 0
```

Attention

Il est important de ne pas confondre les trois types suivants qui représentent des structures de données bien différentes en mémoire :

```
int * int * int
(int * int) * int
int * (int * int)
```

Un n -uplet peut être passé en argument à une fonction ou renvoyé comme résultat. Pour simplifier l'accès à ses composantes, les expressions fonctionnelles sont étendues avec la syntaxe : `fun <motif> -> <expr>`

Exemple

```
1 let f (x, y, (a, b)) = x + y * a - b
```

2 Enregistrements

La structure de n -uplets a deux défauts majeurs.

- Le type produit ne donne pas assez d'informations pour identifier avec précision les objets qu'il représente.
- Un n -uplet avec beaucoup de composantes devient très vite compliqué à utiliser en pratique.

Exemple

Pour une base de données regroupant des informations sur des personnes : nom, prénom, adresse, date de naissance, téléphone fixe, téléphone portable.

```
1 let v =
2     ("Durand", "Jacques",
3      ("2 rue J.Monod", "Orsay Cedex", 91893),
4      (10,03,1967), "0130452637", "0645362738")
```

La consultation des composantes sous cette forme est pénible. De plus, il est très facile de confondre les composantes de ces n -uplets, qui ont le même type, mais qui représentent des informations bien différentes.

2.1 Produits nommés

Pour résoudre ces problèmes, on préfère utiliser des **produits nommés**, aussi appelés **enregistrements** (ou *records* en anglais). Un produit nommé est un n -uplet où les composantes ont chacune un identificateur distinct. Par ailleurs, OCaml exige de donner un nom à chaque produit nommé.

Exemple

```
1 type complex = { re : float ; im : float }
```

La définition ci-dessus définit le type `complex` comme un produit nommé avec deux champs `re` et `im`, chacun de type `float`.

Les produits nommés sont de la forme suivante :

```
1 { <champ_1> : <type_1> ; ... ; <champ_n> : <type_n> }
2 \end{verbatim}
```

où les champs `<champ_i>` ont des noms distincts.

De plus, la règle de portée lexicale pour les champs est la même que pour les déclarations, c'est-à-dire que le nom d'un champ peut masquer un champ d'une définition précédente.

Il y a deux manières de créer des produits nommés.

1. Donner les valeurs de chaque champ en utilisant la syntaxe suivante :

```
1 { <champs_1> = <expr_1> ; ... ; <champs_n> = <expr_n> }
```

Pour que cette expression soit bien typée, il faut que chaque expression `<expr_i>` ait le type `<type_i>` attendu pour le champ `<champ_i>`.

2. Utiliser la notation `with` pour créer un produit nommé à partir d'un autre produit (du même type). La forme générale de cette notation est la suivante :

```
1 { <expr> with <champs_i> = <expr_i> ; ... ; <champs_k> = <expr_k> }
```

L'ordre dans lequel les champs sont donnés n'a pas d'importance.

Exemple

Les deux déclarations suivantes sont équivalentes :

```
1 let c3 = { im = 0.5 ; re = 1.2 }
2 let c2 = { re = 1.2 ; im = 0.5 }
```

Les opérateurs de comparaison sont également insensibles à l'ordre des champs. Ainsi, les deux valeurs `c2` et `c3` ci-dessus seront bien considérées comme étant égales.

Accès aux champs

L'accès aux champs d'un produit nommé peut se faire de deux manières :

— par notation pointée `<expr>.<champ>` ;

```
1 c1.im
```

— par une déclaration avec filtrage par motif.

```
1 type t = { a : int ; b : float * char ; c : string }
2 let v = { a = 1 ; b = (3.4,'a') ; c = "bonjour" }
3
4 let { b = (_,x) ; c = y } = v
```

Comparaison des produits nommés

La relation d'ordre utilisée pour comparer des enregistrements d'un produit nommé *est* l'ordre lexicographique établi selon l'ordre des champs au moment de la définition de `t`.

Exemple

Si on définit

```
1 type t = { b : int ; a : int }
```

et après

```
1 let v1 = { a = 1 ; b = 4 }
2 let v2 = { b = 2 ; a = 2 }
3 let v3 = { b = 3 ; a = 2 }
```

alors on a `v2 < v3 < v1`. Par contre, si on avait défini

```
1 type t = { a : int ; b : int }
```

alors `v1 < v2 < v3`.

La notation avec filtrage par motif est utilisable dans les définitions de fonctions. Ainsi, on peut écrire

```
1 let f { b = (x, _) } = x *. 0.4
```

pour filtrer des valeurs de type `t` en arguments d'une fonction. Par ailleurs, ce motif permet à l'algorithme de typage d'OCaml d'inférer automatiquement que `f` a le type `t -> float`.

Pour comparer n -uplets et enregistrements, on reprend ci-dessous l'exemple des fiches d'une base de donnée.

Exemple

```

1 type address = { street : string ; city : string ; postal_code : int }
2
3 type date = { day : int ; month : int ; year : int }
4
5 type form = {
6   last_name : string ;
7   first_name : string ;
8   address : address ;
9   birthday : date ;
10  phone : string ;
11  mobile : string ;
12 }
```

L'intérêt de cette représentation est qu'elle permet un accès plus simple et sans ambiguïté.

3 Énumérations

Définition

Les *énumérations* sont utilisées pour représenter des ensembles de valeurs appartenant à un domaine fini.

Exemple

Prenons l'exemple d'un jeu de cartes.

Pour représenter les quatre enseignes *Pique*, *Coeur*, *Carreau* ou *Trèfle*, on définit le type `enseigne` :

```
1 type enseigne = Pique | Coeur | Carreau | Trefle
```

Cette définition peut se lire de la manière suivante : « une valeur de type `enseigne` est soit la valeur `Pique`, soit `Coeur`, soit `Carreau` ou bien `Trefle` ».

D'un point de vue ensembliste, le type `enseigne` correspond à un ensemble (fini) qui est défini comme la somme des valeurs qui sont deux à deux disjointes. C'est la raison pour laquelle ce genre de définition est appelée aussi *somme disjointe*.

Les valeurs du type `enseigne` commencent toutes par une **lettre majuscule** : ceci est obligatoire car c'est ce qui permet à OCaml de distinguer une valeur d'un type et des variables (qui commencent toujours par une minuscule).

Ainsi, `Pique` est une constante de type `enseigne`, tout comme `42` est une constante de type `int`.

3.1 Constructeurs

Les constantes `Pique`, `Coeur`, `Carreau` et `Trefle`, utilisées pour construire des valeurs de type `enseigne`, sont appelées **constructeurs** du type `enseigne`.

L'ordre (de gauche à droite) utilisé pour déclarer les constructeurs induit un ordre sur ces valeurs. C'est cette relation d'ordre qui est utilisée par les opérateurs de comparaison prédéfinis d'OCaml.

Les constructeurs d'une énumération peuvent être utilisés dans n'importe quelles expressions ou structures de données.

Exemple

Le type `ens` ci-dessous définit des enregistrements avec un champ `e` qui contient une valeur de type `enseigne`.

```
1 type ens = { e : enseigne ; m : string }
2 let v = { e = Pique ; m = "pique" }
```

Lorsqu'on manipule une valeur appartenant à une somme disjointe, on effectue généralement une analyse par cas pour savoir de quel constructeur il s'agit, puis on réalise un branchement.

```
1 if v.e = Pique then
2   <expr_1>
3 else if v.e = Coeur then
4   <expr_2>
5 else if v.e = Carreau then
6   <expr_3>
7 else
8   <expr_4>
```

Plutôt que cette « cascade » de if-then-else, on préfère écrire cette analyse par cas en utilisant l'instruction de filtrage `match-with` :

```
1 match v.e with
2 | Pique -> <expr_1>
3 | Trefle -> <expr_2>
4 | Coeur -> <expr_3>
5 | Carreau -> <expr_4>
```

La construction `match-with` permet de regrouper des motifs de filtrage sur une même branche.

Motif *ou*

Le filtrage par motif `ou` permette de factoriser des morceaux de code quand un même traitement doit être réalisé pour plusieurs constructeurs.

Exemple

On peut regrouper les cas Pique et Trefle :

```
1 match v.e with
2 | Pique | Trefle -> <expr_1>
3 | Coeur -> <expr_3>
4 | Carreau -> <expr_4>
```

Motif joker

Représenté par le symbole `_`, ce motif représente tous les cas possibles.

Exemple

On peut factoriser tous les cas autres que Pique et Trefle de la manière suivante.

```
1 match v.e with
2 | Pique -> <expr_1>
3 | Trefle -> <expr_2>
4 | _ -> <expr_joker>
```

L'intérêt principal d'utiliser une construction `match-with` plutôt qu'une cascade de `if-then-else` est que le compilateur OCaml effectue une **analyse d'exhaustivité**, c'est-à-dire qu'il vérifie que tous les cas ont été couverts.

Exemple

Par exemple, si on oublie de traiter le constructeur `Coeur` comme ci-dessous :

```
1 match v.e with
2 | Pique -> <expr_1>
3 | Trefle -> <expr_2>
4 | Carreau -> <expr_4>
```

alors le compilateur émet le message d'avertissement suivant :

```
Warning 8 [partial-match]: this pattern-matching is not exhaustive.
Here is an example of a case that is not matched:
Coeur
```

Cela indique que le filtrage n'est pas exhaustif, et qu'il manque le traitement du constructeur `Coeur`.

Le compilateur peut également détecter les cas inutiles dans un filtrage.

Exemple

Dans le code ci-dessous, l'expression `<expr_1_bis>` associée à la deuxième branche pour le motif `Pique` ne sera jamais évaluée puisque ce cas est déjà filtré à la première branche.


```

1 match v.e with
2 | Pique -> <expr_1>
3 | Trefle -> <expr_2>
4 | Coeur -> <expr_3>
5 | Pique -> <expr_1_bis>
6 | Carreau -> <expr_4>

```

Le compilateur détecte cette erreur et émet le message suivant à l'utilisateur :

File "test.ml", line 25, characters 4-9:

```

25 | | Pique -> ()
    ~~~~~

```

Warning 11 [redundant-case]: this match case is unused.

4 Sommes disjointes avec arguments

Les constructeurs d'une somme disjointe peuvent également avoir des arguments.

Exemple

Pour représenter les valeurs de cartes roi, reine, valet, et points (ou petites cartes), on définit le type `carte` suivant :

```

1 type carte = Roi | Reine | Valet | Point of int

```

Ici une carte est soit un `Roi`, soit une `Reine`, soit un `Valet`, soit une carte `Point x`. Le constructeur `Point` est associé à un argument de type `int` : pour construire une carte à l'aide de ce constructeur, il faut l'appliquer à une valeur entière.

```

1 let c = Point 4

```

Filtrage

L'accès aux valeurs associées au constructeur `Point` se fait en utilisant `Point x` dans une construction de filtrage.

Exemple

```

1 match c with
2 | Roi -> <expr_1>
3 | Reine -> <expr_2>
4 | Valet -> <expr_3>
5 | Point x -> <expr_4>

```

Le filtrage de constructeurs avec arguments peut être fait en profondeur.

Exemple

Si on souhaite réaliser un traitement particulier pour les As, on peut écrire :

```
1 match c with
2 | Roi -> <expr_1>
3 | Reine -> <expr_2>
4 | Valet -> <expr_3>
5 | Point 1 -> <expr_as>
6 | Point x -> <expr_autres_points>
```

Le motif joker peut également être utilisé pour filtrer les arguments des constructeurs.

Exemple

Si la valeur des cartes Point n'a pas d'importance

```
1 match c with
2 | Roi -> <expr_1>
3 | Reine -> <expr_2>
4 | Valet -> <expr_3>
5 | Point 1 -> <expr_as>
6 | Point _ -> <expr_autres_points>
```

Le compilateur OCaml est capable de détecter des cas non traités dans un filtrage pour des constructeurs avec arguments.

4.1 Arguments multiples

Il est parfois nécessaire d'associer plusieurs arguments à un constructeur. OCaml dispose d'une syntaxe pour associer directement plusieurs arguments à un constructeur.

- Les coordonnées des points du plan sont représentées avec un type `coord`. Pour construire des points, on utilise le constructeur `Point` avec un seul argument de type `coord`.

```
1 type coord = { x : int ; y : int }
2 type forme = Point of coord
```

- Le constructeur `Cercle` a lui deux arguments, le premier de type `coord` qui contient les coordonnées du centre du cercle, et le second de type `int` qui est le rayon du cercle.

```
1 type coord = { x : int ; y : int }
2 type forme = Cercle of coord * int
```

Exemple

On peut créer un point de coordonnées (10, 10), ainsi qu'un cercle de centre (50, 100) et de rayon 5 de la manière suivante :

```
1 let f1 = Point {x = 10; y = 10}
2 let f2 = Cercle ({x = 50; y = 100}, 5)
```

Filtrage

Pour accéder aux arguments de ces constructeurs, on utilise la construction de filtrage en profondeur.

Exemple

Si `<expr>` est de type `forme`, on peut récupérer toutes les valeurs :

```
1 match <expr> with
2 | Point {x = x; y = y} -> <expr_point>
3 | Cercle ({x = x; y = y}, r) -> <expr_cercle>
```

Les motifs possibles pour filtrer ces valeurs sont les mêmes que ceux utilisés pour les n -uplets. Ainsi, on peut spécifier partiellement les noms des champs d'un enregistrement, utiliser le motif joker, etc.

Exemple

```
1 match <expr> with
2 | Point {x = x} -> <expr_point>
3 | Cercle (p, _) -> <expr_cercle>
```

le motif `{x = x}` ne récupère que la composante x d'un point, tandis que le motif `(p, _)` récupère uniquement l'enregistrement p associé au centre du cercle.